

Correction 2 : Initiation aux éléments finis 1D

Ex 1 *Éléments Finis Unidimensionnels*

1)

La formulation faible

$$\int_0^1 (u' v' + u v) dx = \int_0^1 v dx$$

2)

Le listing complet du programme est donné à la réponse de la question 5).
Pour le calcul de COOR et de CONNEC, il faut lire les lignes 8-24.

3)

Le système élémentaire sur le k ème élément s'écrit :

$$A^k U^k = F^k \quad (1)$$

où :

$$a_{ij}^k = \int_{x_1^k}^{x_2^k} [\psi_j^{k'} \psi_i^{k'} + \psi_j^k \psi_i^k] dx = \int_{-1}^1 \left[\left(\frac{1}{Jac^k} \right)^2 \frac{dL_j}{d\xi} \frac{dL_i}{d\xi} + L_j L_i \right] Jac^k d\xi$$

$$f_i^k = \int_{x_1^k}^{x_2^k} \psi_i^k dx = \int_{-1}^1 L_i Jac^k d\xi$$

où Jac^k est le Jacobien de la transformation

$$Jac^k = \frac{x_2^k - x_1^k}{2}$$

c.f. lignes 30-50 du listing.

4)

c.f. lignes 55-62 du listing.

5)

Listing du programme permettant la résolution avec l'élément P_1 :

```

1 // on charge les programmes dont on a besoin
2 //
3 exec('P1.sci'); exec('gauss.sci');
4 // nbre de points pour l'integration numerique
5 Ng = 3;
6 [XG,PG] = gauss(Ng);
7 //
8 //
9 Nelt = 20; // nbre d'elements
10
11 Nno = 2; // nbre de noeud par element
12
```

```

13 Npt = Nelt * (Nno-1)+1; // nbre de points
14
15 Dx = 1./(Npt-1);
16
17 COOR = matrix((0:Dx:1),Npt,1); // Coordonnees des points
18
19 CONNEC = zeros(Nelt,Nno);
20 for i=1:Nelt
21     for j=1:Nno
22         CONNEC(i,j) = j + (i-1); // matrice de connectivite
23     end;
24 end;
25 //
26 // Fin initialisation des tableaux
27
28 A_glob = zeros(Npt,Npt); F_glob = zeros(Npt,1);
29
30 for iel = 1:Nelt
31     // calcul du systeme elementaire sur chaque element
32     A_el = zeros(Nno,Nno);
33     F_el = zeros(Nno,1);
34
35     x1 = COOR(CONNEC(iel,1)); x2 = COOR(CONNEC(iel,2));
36     jac = (x2 - x1)/2.; // le jacobien est constant sur l'element
37
38     for ig=1:Ng // boucle pour l'integration numerique
39
40         [FP,DxFP] = P1(XG(ig)); // choix des fonctions tests
41
42         DxFP = DxFP/jac;
43
44         for i=1:Nno
45             F_el(i) = F_el(i) + PG(ig)*FP(i)*jac;
46             for j=1:Nno
47                 A_el(i,j) = A_el(i,j) + PG(ig)*(DxFP(j)*DxFP(i) + FP(i)*FP(j))*jac;
48             end;
49         end;
50     end;
51     // fin calcul systeme elementaire pour l'element iel
52
53 // on doit assembler
54
55     for i=1:Nno
56         i_glob = CONNEC(iel,i);
57         F_glob(i_glob) = F_glob(i_glob) + F_el(i);
58         for j=1:Nno
59             j_glob = CONNEC(iel,j);
60             A_glob(i_glob,j_glob) = A_glob(i_glob, j_glob) + A_el(i,j);
61         end;
62     end;
63 end; // fin boucle sur les elements
64
65 //

```

```

66 // On impose les deux conditions aux limites de Diriclet
67 //      U(0) = U(1) = 0
68
69 A = [zeros(1,Npt); A_glob(2:(Npt-1),:); zeros(1,Npt)];
70 A(1,1) = 1; A(Npt,Npt) = 1;
71 F = [0; F_glob(2:(Npt-1)); 0];
72
73 u_cal = A\F;
74
75 // comparaison avec la solution exacte
76 //
77 x = COOR;
78 u_exact = 1- (exp(x) + exp(1-x))/(1+exp(1));
79 erreur = norm(u_cal-u_exact)/norm(u_exact)
80
81 xbasec();
82 plot2d(x,u_cal,style=-1);
83 x = linspace(0,1,100);
84 u_exact = 1- (exp(x) + exp(1-x))/(1+exp(1));
85 plot2d(x,u_exact,style=1);
86 xtitle('Solution P1; Nelt =' + string(Nelt) + ', Npt =' + string(Npt), 'x', 'u');
87

```




FIG. 1 – Solution du problème pour $N_{elt} = 20$: solution exacte (-), solution calculée (+).




FIG. 2 – Evolution de l'erreur $\|u - u_{exact}\|/\|u_{exact}\|$ en fonction de N_{pt}

6)

Si on veut utiliser des éléments \mathbf{P}_2 , les modifications sont mineures par rapport au programme précédent :

- Il faut appeler la fonction qui calcule les polynômes de Lagrange de degré 2 (Ligne 3).
- Modifier la valeur \mathbf{Nno} (Ligne 11).
- Le calcul de la matrice de connectivité (Lignes 20-24).
- L'appel aux fonctions tests L_2 lors de l'intégration numérique (Ligne 40).

```
3   exec('P2.sci'); exec('gauss.sci');

11  Nno = 3; // nbre de noeuds par element

20  for i=1:Nelt
21      CONNEC(i,1) = 1 + 2*(i-1);
22      CONNEC(i,2) = 3 + 2*(i-1);
23      CONNEC(i,3) = 2 + 2*(i-1);
24  end;

40  [FP,DxFP] = P2(XG(ig));
```




FIG. 3 – Evolution de l'erreur $\|u - u_{exact}\|/\|u_{exact}\|$ en fonction de \mathbf{Npt} pour des éléments \mathbf{P}_2 .

7)

les modifications sont les suivantes :

On libère la valeur de u en $x = 1$ aux lignes 69-71.

```
69 A = [zeros(1,Npt); A_glob(2:Npt,:)];
70 A(1,1) = 1;
71 F = [0; F_glob(2:Npt)];
```




FIG. 4 – Solution du problème pour $N_{el} = 10$ avec des éléments \mathbf{P}_2 : solution exacte (-), solution calculée (+).